APPENDIX 4. PERTURBATIONS TO ELECTRON DENSITY MODELS WITH INPUT PARAMETER FORMS

The following perturbations to electron density models (irregularities) are available. The input parameter forms, which describe the perturbation, and the subroutine listings are given on the pages shown.

To add other perturbations to electron density models the user must write a subroutine to modify the normalized electron density (X) and its gradient ($\partial X/\partial r$, $\partial X/\partial \theta$, $\partial X/\partial \varphi$) as a function of position in spherical polar coordinates (r, θ, φ).

The restrictions on electron density models also apply to perturbations. Again, the coordinates r, θ, φ refer to the computational coordinate system, which may not be the same as geographic coordinates. In particular, they are geomagnetic coordinates when the earth-centered dipole model of the earth's magnetic field is used.

The input to the subroutine is through blank common (see Table 3) for the position (r, θ, φ) and through common block /XX/ (see Table 8) for the unperturbed electron density and its gradient. The output is through common block /XX/. It is useful if the name of the subroutine suggests the perturbation model to which it corresponds. It should have an entry point ELECT1 so that it may be called by an electron density subroutine. Any parameters needed by the subroutine should be input into W151 through W199 of the W array. (See Table 2.)

If no perturbation is wanted, the following subroutine should be used.

```
      SUBROUTINE ELECT1                                                 ELEC001
C          USE WHEN AN ELECTRON DENSITY PERTURBATION IS NOT WANTED      ELEC002
      COMMON /XX/ MODX(2),X(6)                                          ELEC003
      COMMON /WW/ ID(10),W0,W(400)                                      ELEC004
      EQUIVALENCE (PERT,W(150))                                         ELEC005
      DATA (MODX(2)=6H NONE )                                           ELEC006
      PERT=0.                                                           ELEC007
      RETURN                                                            ELEC008
         END                                                            ELEC009-
```

INPUT PARAMETER FORM FOR SUBROUTINE TORUS

A perturbation to an ionospheric electron density model consisting of an East-West irregularity with an elliptical cross section above the equator

$$N = N_o (1 + \Delta)$$

$$\Delta = C_o \exp\left\{ - \left[\frac{(R_o + H_o)(\theta - \pi/2) \cos \beta + (R - R_o - H_o) \sin \beta}{A} \right]^2 - \left[\frac{(R - R_o - H_o) \cos \beta - (R_o + H_o)(\theta - \pi/2) \sin \beta}{B} \right]^2 \right\}$$

R_o is the radius of the earth.

R, θ, φ give the position in spherical polar coordinates.

$N_o(R, \theta, \varphi)$ is any ionospheric electron density model.

Specify:

C_o = __________. (W151)

Semi-major axis of ellipse, A = __________km (W152)

Semi-minor axis of ellipse, B = __________km (W153)

Tilt of ellipse, β = __________ degrees (W154)

Height of torus from ground, H_o = __________ km (W155)

(W150: = 1. to use perturbation, = 0. to ignore perturbation)

```
      SUBROUTINE TORUS
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(C0,W(151)),(A,W(152)),(B,W(153)),
     1 (BETA,W(154)),(H0,W(155))
      REAL LAMBDA
      DATA (PDPP=0.),(MODX(2)=6H TORUS)
      ENTRY ELECT1
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURN
      IF (C0.EQ.0.) RETURN
      R0=EARTHR+H0
      Z=R(1)-R0
      LAMBDA=R0*(R(2)-PID2)
      SINBET=SIN(BETA)
      COSBET=COS(BETA)
      P=LAMBDA*COSBET+Z*SINBET
      Y=Z*COSBET-LAMBDA*SINBET
      DELTA=C0*EXP(-(P/A)**2-(Y/B)**2)
      DEL1=DELTA+1.
      PDPR=-2.*DELTA*(P*SINBET/A**2+Y*COSBET/B**2)
      PDPT=-2.*DELTA*(P*R0*COSBET/A**2-Y*R0*SINBET/B**2)
      PXPR=PXPR*DEL1+X*PDPR
      PXPTH=PXPTH*DEL1+X*PDPT
      PXPPH=PXPPH*DEL1+X*PDPP
      X=X*DEL1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE DTORUS

A perturbation to an ionospheric electron density model consisting of two east-west irregularities with elliptical cross sections above the equator. Since the model is expressed in spherical coordinates and does not depend on longitude, the perturbation is actually a torus circling the earth above the equator.

$$N = N_\theta \; (1 + \Delta)$$

$$\Delta = C_1 \exp\left\{ - \left[\frac{(r_\theta + H_1)(\theta - \pi/2)\cos\beta + (r - r_\theta - H_1)\sin\beta}{A_1} \right]^2 - \left[\frac{(r - r_o - H_1)\cos\beta - (r_\theta + H_1)(\theta - \pi/2)\sin\beta}{B_1} \right]^2 \right\}$$

$$+ C_2 \exp\left\{ - \left[\frac{(r_\theta + H_2)(\theta - \pi/2 + \delta\theta)\cos\beta + (r - r_\theta - H_2)\sin\beta}{A_2} \right] - \left[\frac{(r - r_\theta - H_2)\cos\beta - (r_\theta + H_2)(\theta - \pi/2 + \delta\theta)\sin\beta}{B_2} \right]^2 \right\}$$

$\delta\theta$ = Northward angular displacement of the lower blob from the upper one

$$= \frac{H_1 - H_2}{\tan\beta \; (r_o + H_2)}$$

r_θ is the radius of the earth.

r, θ, φ are spherical (earth-centered) polar coordinates.

$N_o(r, \theta, \varphi)$ is any electron density model.

Specify:

use perturbation____________(W150 = 1.)
ignore perturbation____________(W150 = 0.)

Fractional perturbation electron density at the center of the upper blob, C_1 = ____________(W151).

That of the lower blob, C_2 = ____________(W156).

Height (above ground) of the center of the upper blob, H_1 = ____________km (W155).

That of the lower blob, H_2 = ______________ km (W159).

Angle (with a horizontal southward vector) of the line joining the blob centers, θ = ______________ rad/deg (W154).

Semi-axis of the upper blob, to the 1/e perturbation contour, in the direction of the line joining the blobs, A_1 = ____________ km (W152).

That of the lower blob, A_2 = ______________ km (W157).

Semi-axis of the upper blob in the direction normal to the line joining the blobs, B_1 = ____________ km (W153).

That of the lower blob, B_2 = ____________ km (W158).

```
      SUBROUTINE DTORUS
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT
      COMMON R(6) /WW/ ID(10),WO,W(400)
      EQUIVALENCE (EARTHR,W(2)),(C1,W(151)),(A1,W(152)),(B1,W(153)),
     1 (BETA,W(154)),(H1,W(155)),(C2,W(156)),(A2,W(157)),(B2,W(158)),
     2 (H2,W(159))
      REAL LAMBDA1,LAMBDA2
      DATA (MODX(2)=6HDTORUS),(PDPP=0.)
      ENTRY ELECT1
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURN
      IF (C1.EQ.0.) RETURN
      R1=EARTHR+H1
      R2=EARTHR+H2
      Z1=R(1)-R1
      Z2=R(1)-R2
      LAMBDA1=R1*(R(2)-PID2)
      LAMBDA2=R2*(R(2)-PID2+(H1-H2)/R2/TANF(BETA))
      SINBET=SIN(BETA)
      COSBET=COS(BETA)
      P1=LAMBDA1*COSBET+Z1*SINBET
      P2=LAMBDA2*COSBET+Z2*SINBET
      Y1=Z1*COSBET-LAMBDA1*SINBET
      Y2=Z2*COSBET-LAMBDA2*SINBET
      DELTA1=C1*EXP(-(P1/A1)**2-(Y1/B1)**2)
      DELTA2=C2*EXP(-(P2/A2)**2-(Y2/B2)**2)
      DEL1=1.+DELTA1+DELTA2
      PDPR1=-2.*DELTA1*(P1*SINBET/A1**2+Y1*COSBET/B1**2)
      PDPR2=-2.*DELTA2*(P2*SINBET/A2**2+Y2*COSBET/B2**2)
      PDPT1=-2.*DELTA1*(P1*R1*COSBET/A1**2-Y1*R1*SINBET/B1**2)
      PDPT2=-2.*DELTA2*(P2*R2*COSBET/A2**2-Y2*R2*SINBET/B2**2)
      PXPR=PXPR*DEL1+X*(PDPR1+PDPR2)
      PXPTH=PXPTH*DEL1+X*(PDPT1+PDPT2)
      PXPPH=PXPPH*DEL1*PDPP
      X=X*DEL1
      RETURN
      END
```

INPUT PARAMETER FORM FOR SUBROUTINE TROUGH

A perturbation to an ionospheric electron density model consisting of an increase in electron density near any latitude

$N = (1 + \Delta)\, N_0\ (R,\ \theta,\ \varphi)$ $\qquad W = B$ for $\frac{\pi}{2} - \theta - \lambda \geq 0$

$\Delta = A \exp\left(-\left(\frac{\pi/2 - \theta - \lambda}{W}\right)^2\right)$ $\qquad W = B \times C$ for $\frac{\pi}{2} - \theta - \lambda < 0$

$N_0\ (R,\ \theta,\ \varphi)$ is any ionospheric electron density model.

$R,\ \theta,\ \varphi$ give the position in spherical polar coordinates.

Specify:

Amplitude of the perturbation, A = __________ (W151)

half width of the perturbation, B = __________ degrees (W152)

latitude of the perturbation, λ = __________ degrees (W153)

width factor for South of trough, C = __________ (W154)

(W150: = 1. to use perturbation, = 0. to ignore perturbation)

```
      SUBROUTINE TROUGH
C             A PERTURBATION TO AN ELECTRON DENSITY MODEL
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),WO,W(400)
      EQUIVALENCE (A,W(151)),(E,W(152)),(ALAT,W(153)),(FACTOR,W(154))
      DATA (MODX(2)=6HTROUGH)
      ENTRY ELECT1
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURN
      IF (A.EQ.0.) RETURN
      ANGLE=R(2)+ALAT-PID2
      WIDTH=B
      IF (ANGLE.GT.0.)    WIDTH=FACTOR*B
      ANGLE=ANGLE/WIDTH
      DELTA=A*EXP(-ANGLE**2)
      DEL1=DELTA+1.
      PXPR=PXPR*DEL1
      PXPTH=PXPTH*DEL1-2.*X*ANGLE*DELTA/WIDTH
      PXPPH=PXPPH*DEL1
      X=X*DEL1
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE SHOCK

A perturbation to an ionospheric electron density model consisting of an increase in electron density produced by a shock wave

$$N(R, \theta, \varphi) = N_o(R, \theta, \varphi)\left[1 + P \exp\left(-9\left(\frac{\rho_c - \rho}{w}\right)^2\right)\right]$$

$$\rho_c = s(h - h_o) - w$$

$$\rho = R\left|\cos^{-1}\left[\cos(\varphi - \varphi_o)\cos(\lambda - \lambda_o)\right]\right|$$

$N_o(R, \theta, \varphi)$ is the ambient electron density specified by any electron density model.

R, θ, φ give the position in spherical polar coordinates.

$h = R - a$ is the height above the surface of the earth.

a is the radius of the earth.

$\lambda = \frac{\pi}{2} - \theta$ is the latitude.

Specify:

Relative increase in electron density, $P =$ ________ (W151).

Width of the disturbance, $w =$ ________ km (W152).

Latitude of the center of the disturbance, $\lambda_o =$ ________ radians or degrees (W153).

Longitude of the center of the disturbance, $\varphi_o =$ ________ radians or degrees (W154).

Slope measured from vertical - rate of increase of ρ_c with height, $s =$ ________ (W155).

Height to the bottom of the disturbance, $h_o =$ ________ km (W156).

(W150: = 1. to use perturbation, = 0. to ignore perturbation)

```
      SUBROUTINE SHOCK
C    A PERTURBATION TO AN ELECTRON DENSITY MODEL SIMULATING A SHOCK WAVE
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX
      COMMON R(6) /WW/ ID(10),W0,W(400)
      EQUIVALENCE (EARTHR,W(2)),(P,W(151)),(WW,W(152)),(ALAT,W(153)),
     1 (ALON,W(154)),(S,W(155)),(H0,W(156))
      REAL LAT,LON
      DATA (MODX(2)=6H SHOCK)
      ENTRY ELECT1
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURN
      IF (P.EQ.0..OR.WW.EQ.0.)  RETURN
      H=R(1)-EARTHR
      RHOC=S*(H-H0)-WW
      LON=R(3)-ALON
      LAT=PID2-R(2)-ALAT
      COSLON=COS(LON)
      COSLAT=COS(LAT)
      U=COSLON*COSLAT
      RHO=R(1)*ACOS(U)
      DIF=RHOC-RHO
      CON=-9./WW**2
      CONS=P*EXP(CON*DIF**2)
      CONST=1.+CONS
      CON=2.*CON*CONS*DIF
      PXPR=PXPR*CONST+X*CON*(S-RHO/R(1))
      CONS=R(1)*(1./SQRT(1.-U**2))
      PXPTH=PXPTH*CONST+X*CON*CONS*COSLON*SIN(LAT)
      PXPPH=PXPPH*CONST-X*CON*CONS*COSLAT*SIN(LON)
      X=X*CONST
      RETURN
         END
```

INPUT PARAMETER FORM FOR SUBROUTINE WAVE

A perturbation to an ionospheric electron density model consisting of a "gravity-wave" irregularity traveling from north pole to south pole

$$N = N_0(1 + \Delta)$$

$$\Delta = \delta \exp\left\{- [(R - R_0 - z_0)/H]^2\right\} \cdot \cos\left\{2\pi \left[t' + (\pi/2 - \theta) \frac{R_0}{\lambda_x} + (R - R_0)/\lambda_z\right]\right\}$$

$$\frac{\partial N}{\partial t} = \frac{-2\pi}{\lambda_x} V_x N_0 \delta \exp - [(R - R_0 - z_0)/H]^2 \cdot \sin 2\pi \left[t' + (\pi/2 - \theta) \frac{R_0}{\lambda_x} + (R - R_0)/\lambda z\right]$$

R_0 is the radius of the earth.

R, θ, φ are the spherical (earth-centered) polar coordinates (Δ is independent of φ).

N_0 (R, θ, φ) is any electron density model.

Specify:

the height of maximum wave amplitude, z_0 = ______km (W151)

wave-amplitude "scale height," H = ______km (W152)

wave perturbation amplitude, δ = ______ [0. to 1.] (W153)

horizontal trace velocity, V_x = ______km/sec (W154)
(needed only if Doppler shift is calculated)

horizontal wavelength, λ_x = ______km (W155)

vertical wavelength, λ_z = ______km (W156)

time in wave periods, t' = ______ [0. to 1.] (W157)

(W150: = 1. to use perturbation, = 0. to ignore perturbation)

```
      SUBROUTINE WAVE                                                   WAVE001
C          PERTURBATION TO AN ALPHA-CHAPMAN ELECTRON DENSITY MODEL      WAVE002
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)                                WAVE003
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT,HMAX                  WAVE004
      COMMON R(6) /WW/ ID(10),W0,W(400)                                 WAVE005
      EQUIVALENCE (EARTHR,W(2)),(Z0,W(151)),(SH,W(152)),(DELTA,W(153)), WAVE006
     1 (VSUBX,W(154)),(LAMBDAX,W(155)),(LAMBDAZ,W(156)),(TP,W(157))     WAVE007
      REAL LAMBDAX,LAMBDAZ                                              WAVE008
      DATA (MODX(2)=6H WAVE )                                           WAVE009
      ENTRY ELECT1                                                      WAVE010
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURNWAVE011
      IF (DELTA.EQ.0..OR.SH.EQ.0.)    RETURN                            WAVE012
      H=R(1)-EARTHR                                                     WAVE013
      EXPQ=EXP(-((H-Z0)/SH)**2)                                         WAVE014
      TMP=PIT2*(TP+(PID2-R(2))*EARTHR/LAMBDAX+H/LAMBDAZ)                WAVE015
      SINW=SIN(TMP)                                                     WAVE016
      COSW=SIN(PID2-TMP)                                                WAVE017
      CONS=1.0+DELTA*EXPQ*COSW                                          WAVE018
      IF (H.NE.0.) PXPR=PXPR*CONS-X*DELTA*EXPQ*(2.0/SH**2*(H-Z0)*COSW   WAVE019
     1 +PIT2/LAMBDAZ*SINW)                                              WAVE020
      PXPTH=PXPTH*CONS+X*DELTA*PIT2*EARTHR/LAMBDAX*SINW*EXPQ            WAVE021
      PXPPH=PXPPH*CONS                                                  WAVE022
      PXPT=0.                                                           WAVE023
      IF (VSUBX.NE.0.) PXPT=-PIT2*VSUBX/LAMBDAX*X*DELTA*EXPQ*SINW       WAVE024
      X=X*CONS                                                          WAVE025
      RETURN                                                            WAVE026
         END                                                            WAVE027-
```

INPUT PARAMETER FORM FOR SUBROUTINE WAVE 2

PERTURBATION TO AN IONOSPHERIC ELECTRON DENSITY MODEL

A "gravity-wave" irregularity traveling from north pole to south pole - same as WAVE 1, but with Gaussian amplitude variations in latitude and longitude, and provision for a horizontal "group velocity"

$N = N_0 \ (1 + AC)$

$$A = \delta \exp - \left(\frac{r - r_0 - z_0}{H} \right)^2 \cdot \exp - \left(\frac{\theta - \theta_0(t)}{\Theta} \right)^2 \cdot \exp - \left(\frac{\varphi}{\Phi} \right)^2$$

$$C = \cos 2\pi \left[t' + (\pi/2 - \theta) \frac{r_0}{\lambda_x} + (r - r_0)/\lambda_z \right]$$

$$\theta_0 = \theta_\infty + V_g t / r_0$$

r_0 is the radius of the earth.

r, θ, φ are spherical (earth-centered) polar coordinates.

$N_0(r, \theta, \varphi)$ is any electron density model.

Specify:

use perturbation____________(W150 = 1.)

ignore perturbation____________(W150 = 0.)

the height of maximum wave amplitude, z_0 = __________km (W151)

wave-amplitude "scale height," H = __________km (W152)

wave perturbation amplitude, δ = __________(0 to 1) (W153)

horizontal trace velocity, V_x = __________km/sec. (W154)
(needed only if Doppler shift is calculated)

horizontal wavelength, λ_x = __________km (W155)

vertical wavelength, λ_z = __________km (W156)

time in wave periods, t' = __________ (W157)

amplitude "scale distance" in latitude, Θ = __________degrees (W159)

amplitude "scale distance" in longitude, Φ = __________degrees (W160)

latitude of maximum amplitude at t = 0, θ_∞ = __________degrees (W158)

southward group velocity, V_g = __________km/sec (W161)
(needed even if Doppler shift is not calculated)

```
      SUBROUTINE WAVE2                                                  WAV2001
C          PERTURBATION TO AN ANY ELECTRON DENSITY MODEL               WAV2002
      COMMON /CONST/ PI,PIT2,PID2,DUM(5)                                WAV2003
      COMMON /XX/ MODX(2),X,PXPR,PXPTH,PXPPH,PXPT                       WAV2004
      COMMON R(6) /WW/ ID(10),W0,W(400)                                 WAV2005
      EQUIVALENCE (EARTHR,W(2)),(Z0,W(151)),(SH,W(152)),(DELTA,W(153)), WAV2006
     1 (VSUBX,W(154)),(LAMBDAX,W(155)),(LAMBDAZ,W(156)),(TP,W(157)),    WAV2007
     2 (TH00,W(158)),(THC,W(159)),(PHIC,W(160)),(VGX,W(161))            WAV2008
      REAL LAMBDAX,LAMBDAZ                                              WAV2009
      DATA (MODX(2)=6H WAVE2)                                           WAV2010
      ENTRY ELECT1                                                      WAV2011
      IF (X.EQ.0..AND.PXPR.EQ.0..AND.PXPTH.EQ.0..AND.PXPPH.EQ.0.) RETURNWAV2012
      IF (DELTA.EQ.0..OR.SH.EQ.0.)  RETURN                              WAV2013
      H=R(1)-EARTHR                                                     WAV2014
      TH0=TH00+LAMBDAX*TP*VGX/VSUBX/EARTHR                              WAV2015
      EXPR=EXP(-((H-Z0)/SH)**2)                                         WAV2016
      EXPTH=EXP(-((R(2)-TH0)/THC)**2)                                   WAV2017
      EXPPHI=EXP(-(R(3)/PHIC)**2)                                       WAV2018
      WW=PIT2*(TP+(PID2-R(2))*EARTHR/LAMBDAX+H/LAMBDAZ)                 WAV2019
      SINW=SIN(WW)                                                      WAV2020
      COSW=COS(WW)                                                      WAV2021
      E=DELTA*EXPR*EXPTH*EXPPHI                                         WAV2022
      CONS=1.0+E*COSW                                                   WAV2023
      PXPR=PXPR*CONS-X*E*2.*(COSW*(H-Z0)/SH**2+PI/LAMBDAZ*SINW)         WAV2024
      PXPTH=PXPTH*CONS+2.*E*(X*PI*EARTHR*SINW/LAMBDAX-(R(2)-TH0)/       WAV2025
     1 THC**2*COSW)                                                     WAV2026
      PXPPH=PXPPH*CONS-X*2.*E*R(3)/PHIC**2*COSW                         WAV2027
      PXPT=-PIT2*VSUBX*E/LAMBDAX*SINW+2.0*E*VGX/EARTHR*COSW*(R(2)-TH0   WAV2028
     1 -LAMBDAX*TP/EARTHR)/THC                                          WAV2029
      X=X*CONS                                                          WAV2030
      RETURN                                                            WAV2031
         END                                                            WAV2032-
```

INPUT PARAMETER FORM FOR SUBROUTINE DOPPLER

HEIGHT PROFILE OF dN/dt

(A perturbation to an ionospheric electron density model which calculates the time derivative of electron density for calculating Doppler shifts)

First card tells how many profile points in I4 format. The cards following the first card give the height and dN/dt of the profile points one point per card in F8. 2, E12. 4 format. The heights must be in increasing order. Set W151 =1. 0 to read in a new profile. After the cards are read, DOPPLER will reset W151=0. This subroutine makes an exponential extrapolation down using the bottom 2 points in the profile.

1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20	1 2 3 4 5 6 7 8	9 10 11 12 13 14 15 16 17 18 19 20
HEIGHT h km	TIME DERIVATIVE OF ELECTRON DENSITY dN/dt ELECTRONS/cm³/sec	HEIGHT h km	TIME DERIVATIVE OF ELECTRON DENSITY dN/dt ELECTRONS/cm³/sec

```
      SUBROUTINE DOPPLER                                                DOPP001
C       COMPUTES DN/DT                 FROM PROFILES HAVING THE SAME FORMDOPP002
C       AS THOSE USED BY SUBROUTINE TABLE X                             DOPP003
C MAKES AN EXPONENTIAL EXTRAPOLATION DOWN USING THE BOTTOM TWO POINTS   DOPP004
C         NEEDS SUBROUTINE GAUSEL                                       DOPP005
      DIMENSION HPC(250),FN2C(250),ALPHA(250),BETA(250),GAMMA(250),     DOPP006
     1 DELTA(250),SLOPE(250),MAT(4,5)                                   DOPP007
      COMMON /CONST/ PI,PIT2,PID2,DEGS,RAD,K,DUM(2)                     DOPP008
      COMMON /XX/ MODX(2),XDUM,PXPR,PXPTH,PXPPH,X,HMAX                  DOPP009
      COMMON R(6) /WW/ ID(10),W0,W(400)                                 DOPP010
      EQUIVALENCE (EARTHR,W(2)),(F,W(6)),(READFN,W(151))                DOPP011
      REAL MAT,K                                                        DOPP012
      DATA (MODX(2)=7HDOPPLER)                                          DOPP013
      ENTRY ELECT1                                                      DOPP014
      IF (READFN.EQ.0.) GO TO 10                                        DOPP015
      READFN=0.                                                         DOPP016
      READ 1000, NOC,(HPC(I),FN2C(I),I=1,NOC)                           DOPP017
 1000 FORMAT (I4/(F8.2,E12.4))                                          DOPP018
      PRINT 1200, (HPC(I),FN2C(I),I=1,NOC)                              DOPP019
 1200 FORMAT(1H1,14X,6HHEIGHT,4X,16H      DN/DT       /(1X,F20.10,E20.10))DOPP020
      A=0.                                                              DOPP021
      IF(FN2C(1).NE.0.) A=ALOG(FN2C(2)/FN2C(1))/(HPC(2)-HPC(1))         DOPP022
      FN2C(1)=K*FN2C(1)                                                 DOPP023
      FN2C(2)=K*FN2C(2)                                                 DOPP024
      SLOPE(1)=A*FN2C(1)                                                DOPP025
      SLOPE(NOC)=0.                                                     DOPP026
      NMAX=1                                                            DOPP027
      DO 6 I=2,NOC                                                      DOPP028
      IF (FN2C(I).GT.FN2C(NMAX)) NMAX=I                                 DOPP029
      IF (I.EQ.NOC) GO TO 4                                             DOPP030
      FN2C(I+1)=K*FN2C(I+1)                                             DOPP031
      DO 3 J=1,3                                                        DOPP032
      M=I+J-2                                                           DOPP033
      MAT(J,1)=1.                                                       DOPP034
      MAT(J,2)=HPC(M)                                                   DOPP035
      MAT(J,3)=HPC(M)**2                                                DOPP036
    3 MAT(J,4)=FN2C(M)                                                  DOPP037
      CALL GAUSEL (MAT,4,3,4,NRANK)                                     DOPP038
      IF (NRANK.LT.3) GO TO 60                                          DOPP039
      SLOPE(I)=MAT(2,4)+2.*MAT(3,4)*HPC(I)                              DOPP040
    4 DO 5 J=1,2                                                        DOPP041
      M=I+J-2                                                           DOPP042
      MAT(J,1)=1.                                                       DOPP043
      MAT(J,2)=HPC(M)                                                   DOPP044
      MAT(J,3)=HPC(M)**2                                                DOPP045
      MAT(J,4)=HPC(M)**3                                                DOPP046
      MAT(J,5)=FN2C(M)                                                  DOPP047
      L=J+2                                                             DOPP048
      MAT(L,1)=0.                                                       DOPP049
      MAT(L,2)=1.                                                       DOPP050
      MAT(L,3)=2.*HPC(M)                                                DOPP051
      MAT(L,4)=3.*HPC(M)**2                                             DOPP052
    5 MAT(L,5)=SLOPE(M)                                                 DOPP053
      CALL GAUSEL (MAT,4,4,5,NRANK)                                     DOPP054
      IF (NRANK.LT.4) GO TO 60                                          DOPP055
      ALPHA(I)=MAT(1,5)                                                 DOPP056
      BETA(I)=MAT(2,5)                                                  DOPP057
      GAMMA(I)=MAT(3,5)                                                 DOPP058
    6 DELTA(I)=MAT(4,5)                                                 DOPP059
      HMAX=AMAX1(HMAX,HPC(NMAX))                                        DOPP060
      NH=2                                                              DOPP061
   10 H=R(1)-EARTHR                                                     DOPP062
      F2=F*F                                                            DOPP063
      IF (H.GE.HPC(1)) GO TO 12                                         DOPP064
   11 NH=2                                                              DOPP065
```

```
                                                                       DOPP066
      IF(FN2C(1).EQ.0.) GO TO 50                                       DOPP067
      X=FN2C(1)*EXP(A*(H-HPC(1)))/F2                                   DOPP068
      GO TO 50                                                         DOPP069
   12 IF (H.GE.HPC(NOC)) GO TO 18                                      DOPP070
      NSTEP=1                                                          DOPP071
      IF (H.LT.HPC(NH-1)) NSTEP=-1                                     DOPP072
   15 IF (HPC(NH-1).LE.H.AND.H.LT.HPC(NH)) GO TO 16                    DOPP073
      NH=NH+NSTEP                                                      DOPP074
      GO TO 15                                                         DOPP075
   16 X=(ALPHA(NH)+H*(BETA(NH)+H*(GAMMA(NH)+H*DELTA(NH))))/F2          DOPP076
      GO TO 50                                                         DOPP077
   18 X=FN2C(NOC)/F2                                                   DOPP078
   50 CONTINUE                                                         DOPP079
      RETURN                                                           DOPP080
   60 PRINT 6000, I,HPC(I)                                             DOPP081
 6000 FORMAT(4H THE,I4,55HTH POINT IN THE   DN/DT          PROFILE HAS TDOPP082
     1HE HEIGHT,F8.2,40H KM, WHICH IS THE SAME AS ANOTHER POINT.)      DOPP083
      CALL EXIT                                                        DOPP084
         END                                                           DOPP085
```